
概要

生産性を表すパラメータを導入したモデルをMPSGEのコードで記述する方法。

1 準備

以下の議論は CES関数のケースにも適用できるが、ここでは例として Cobb-Douglas型生産関
数を利用して議論を進める。

q = ϕ ∏
i

[λiyi]
αi (1)

qは生産量、yi は投入物 iの投入量、ϕ、αi (∑i αi = 1)は定数である。また、λi は投入物 iに関す
る生産性パラメーターであり、値が大きくなるほど生産性が高いことになる。この λi の初期値は

1とする。
生産関数が (1)式で与えられる場合、単位費用関数、単位需要関数はそれぞれ以下のような形式
となる。

c =
1
ϕ ∏

i

[
pi

λiαi

]αi

(2)

xi =
αi
pi

c (3)

pi は投入物 iの価格である。
さらに、以上の関数を、ϕ、αi を消した calibrated share formで書き直すと、

q = q̄ ∏
i

[
xi

x̄i/λi

]θi

(4)

c = c̄ ∏
i

[
pi

λi p̄i

]θi

(5)

xi =
θi
pi

c (6)

となる1。“-”付きの変数はその変数のベンチマーク値を表す。また、θi は benchmarkにおける投
入シェアであり

θi ≡
p̄i x̄i

∑j p̄j x̄j
(7)

1 (3)式、及びベンチマークの p̄i、x̄i、c̄より αi がカリブレートできる。

αi =
p̄i x̄i

c̄
=

p̄i x̄i

∑j p̄j x̄j
= θi

である。

2 MPSGEのコード

以上の準備をもとに (1)のMPSGEでの表現を考える。これまでの記号をMPSGEコードでは次
のように表現する。

q → q

c → c

pi → p(i)

λi → lambda(i)

q̄ → q0

x̄i → x0(i)

p̄i → p0(i)

まず、技術進歩を全く考慮しないケースでは¶ ³
$prod:q s:1

o:c q:q0

i:p(i) q:x0(i) p:p0(i)µ ´
のような形式になる。ここに生産性パラメータ lambda(i)を導入するのであるが、(4)式を見ると
生産性パラメータはベンチマークの生産量を割る形で入ってきているので、以下のように書き換え

ればよいのではないかと考えられる2。

[Case 1]¶ ³
$prod:q s:1

o:c q:q0

i:p(i) q:(x0(i)/lambda(i)) p:p0(i)µ ´
この記述は一見もっともらしく感じるが、実はこれでは技術進歩を正確に導入したことにはなら

ない。その理由はMPSGEでのパラメータのカリブレーション方法にある。MPSGEでは投入物の
reference quantityと priceを利用してベンチマークにおける投入シェアを計算する。そして、(1)
式の αi はその投入シェアによってカリブレートされることになる。

ここで、上のようにMPSGEコードを記述したとき、投入シェアがどう計算されるか求めてみ
ると

αi = θi =
p̄i x̄i/λi

∑j p̄j x̄j/λj
(8)

また、ϕは

ϕ = q̄/ ∏
i

(x̄i)θi

でカリブレートできる。
2lambda(i)は値が大きいほど生産性が高くなるように定義されているので、ベンチマークの生産量を割る形式で導入さ

れる。

となる。この式から明かなように、この場合ベンチマークの投入シェアが λi の値に依存すること

になる3。これは生産性パラメータを変化させることで (1)式の αi をも変化させてしまうことを意

味する。これでは純粋に生産性の変化だけを導入したことにはならない。

以上のように、MPSGEのカリブレーション方法では reference quantity側だけに生産性パラメー
タを導入してしまうと、生産性の変化だけではなく、シェアパラメータの αi までも変化させるこ

とになるという問題が生じる。

それでは、純粋に生産性の変化だけを導入する際にはどのようにコードを記述すべきかというと、

そのヒントは calibrated share formの単位費用関数 (5)式に含まれている。(5)では reference price
にあたる部分が p̄iλi に置き換えられている。これと同じように、MPSGEのコードでも reference
priceの部分を書き換えればよいのである。すなわち

[Case 2]¶ ³
$prod:q s:1

o:c q:q0

i:p(i) q:(x0(i)/lambda(i)) p:(p0(i)*lambda(i))µ ´
とすればよい。

このようにすれば reference quantity、priceから計算される投入シェアは

αi = θi =
( p̄iλi)(x̄i/λi)

∑j( p̄jλj)(x̄j/λj)
=

p̄i x̄i

∑j p̄j x̄j
(9)

となり、λi には依存しなくなるため、αi に影響を与えることなく、生産性パラメータの変化の効果

のみをとりだすことができる。

3 MPSGEの sample code

$title A sample MPSGE program including technological change:

$ontext

Time-stamp: <2009-12-01 21:28:14 Shiro Takeda>

First-written: <2006/07/17>

[Two experiments]

Case 1 -> Reference quantity の生産性パラメータを上昇

Case 2 -> Reference quantity & price の生産性パラメータを上昇

Case 2 が適切に技術進歩を考慮したケース。

[Results]

Case 1 と case 2 で生産量の伸び率が異なってくる。

Case 1 の場合、Cobb-Doublas 型関数であるのに、要素の投入シェアが変わってしまう。

3λi が全て等しいという特殊ケースは除く。

$offtext

* Set definition:

set f Index of factor / k, l /;

display f;

* Parameter definition:

parameter

q0 Reference output

x0(f) Reference input

p0 Reference price of output

pf0(f) Reference price of factors

end0(f) Endowment of factors

lambda_q(f) Efficiency parameter

lambda_p(f) Efficiency parameter

;

q0 = 100;

x0("k") = 0.3 * q0;

x0("l") = q0 - x0("k");

p0 = 1;

pf0(f) = 1;

end0(f) = x0(f);

lambda_q(f) = 1;

lambda_p(f) = 1;

display q0, x0, p0, pf0, end0, lambda_q, lambda_p;

* ----------------------------------------------------------------------

* MPSGE code:

$ontext

$model:sample

$sectors:

q ! Output

$commodities:

p ! Output price

pf(f) ! Price of primary factor

$consumers:

ra ! Representative agent

$prod:q s:1

o:p q:q0

i:pf(f) q:(x0(f)/lambda_q(f)) p:(pf0(f)*lambda_p(f))

$report:

v:x(f) i:pf(f) prod:q ! Quantity of input

$demand:ra

d:p q:q0

e:pf(f) q:end0(f)

$offtext

$sysinclude mpsgeset sample

parameter

result Compare two results;

* ----------------------------------------------------------------------

* Benchmark replication.

sample.iterlim = 0;

$include sample.gen

solve sample using mcp;

result("Bench","q") = q.l;

result("Bench","Sh_k") = pf.l("k") * x.l("k") / (sum(f, pf.l(f) * x.l(f)));

result("Bench","Sh_l") = pf.l("l") * x.l("l") / (sum(f, pf.l(f) * x.l(f)));

result("Bench","v_k") = pf.l("k") * x.l("k") / p.l;

result("Bench","v_l") = pf.l("l") * x.l("l") / p.l;

sample.iterlim = 9000;

* ----------------------------------------------------------------------

* Increase only in lambda_q:

lambda_q("k") = 2;

$include sample.gen

solve sample using mcp;

result("Case1","q") = q.l;

result("Case1","Sh_k") = pf.l("k") * x.l("k") / (sum(f, pf.l(f) * x.l(f)));

result("Case1","Sh_l") = pf.l("l") * x.l("l") / (sum(f, pf.l(f) * x.l(f)));

result("Case1","v_k") = pf.l("k") * x.l("k") / p.l;

result("Case1","v_l") = pf.l("l") * x.l("l") / p.l;

* ----------------------------------------------------------------------

* Increase both in lambda_q and lambda_p:

lambda_q("k") = 2;

lambda_p("k") = 2;

sample.iterlim = 9000;

$include sample.gen

solve sample using mcp;

result("Case2","q") = q.l;

result("Case2","Sh_k") = pf.l("k") * x.l("k") / (sum(f, pf.l(f) * x.l(f)));

result("Case2","Sh_l") = pf.l("l") * x.l("l") / (sum(f, pf.l(f) * x.l(f)));

result("Case2","v_k") = pf.l("k") * x.l("k") / p.l;

result("Case2","v_l") = pf.l("l") * x.l("l") / p.l;

display result;

* --------------------

* Local Variables:

* mode: gams

* fill-column: 90

* End:

---- 448 PARAMETER result Compare two results

q Sh_k Sh_l v_k v_l

Bench 1.000 0.300 0.700 30.000 70.000

Case1 1.130 0.176 0.824 19.943 93.068

Case2 1.231 0.300 0.700 36.934 86.180

